
Overview

The intent of this paper is to walk the developer through the process of creating a VFP search form using the COMCodebook framework. This paper should also familiarize the developer with the mechanics of submitting and retrieving search criteria so they are able to create their own customized search forms. The example developed in this paper is a form to search for consultants in the TimeTrac database. The TimeTrac application is a demonstration application that comes with COMCodebook. This paper assumes you have this application installed and are able to run it.

Step 1: Open Up The TimeTrac Projects

If you do not have the four projects associated with the TimeTrac application open already go ahead and open them. You can either open them manually or by running the startcb.prg found in TimeTrac’s interface directory.

Step 2: Adding A Search Method To The ConsultantDataSource Class

In the next three steps, we will be adding customized search code to TimeTrac’s consultant classes. This involves modifying the interface layer of the ResourceManager, ResourceManagerProxy and DataManager classes. If you are unfamiliar with these classes and their use, please refer to the following white papers: Creating Flexible Searches In COMCodebook DataSources and Developing Apps With COMCodebook.

You will find when developing COMCodebook applications that the majority of the behind-the-scenes work is done within the ResourceManagerProxy, ResourceManager, and DataManager DataSource classes. These are the workhorse classes of COMCodebook. You will also find that due to the relationship these classes have with one another adding code in one class usually involves adding code to the other two classes. If you are familiar with Codebook, this happens in much the same way that the navigation buttons, the bizobjform, and the bizobj classes interact with each other; clicking on the "next" button on the navigation toolbar class calls that buttons Next() method which in turn calls the bizobjform’s Next() method and that calls the bizobj’s Next() method. In object-oriented terms this is called polymorphism. Similarly, a method in the ResourceManagerProxy class will usually make the same method call to the ResourceManager class, which will then call the method in the DataManager class. Keeping this in mind, I find it easiest to code from the bottom up, that is, add code to the DataManager class, then the ResourceManager class, and finally the ResourceManagerProxy class. Other people prefer just the opposite. The approach you take is not important as long as you are consistent. Let’s go ahead and start coding!

Modify the adatasources.prg found in the DataSource project. Scroll down until you find the class called ConsultantDataSource and add the following method and code to the end of that class:

*---------------------------------------
FUNCTION GetConsultantBySearchCriteria( tuFilter )
*---------------------------------------
LOCAL loADO, lcSelect
lcSelect = ""
loADO = THIS.GetADOAggregateParameter()
loADO.oCommand.ActiveConnection = loADO.oConnection
lcSelect = [SELECT Consultants.* FROM Consultants ]
IF THIS.CreateFilterFromResource( tuFilter ) = FILE_OK
WITH THIS.oSearchClauses
loADO.oCommand.CommandText = lcSelect + .cWhere + .cOrder
ENDWITH
ELSE
loADO.oCommand.CommandText = lcSelect + [ WHERE 1 = 0 ]
ENDIF
THIS.oSearchClauses = NULL
THIS.ExecuteSQLQuery( loADO )
RETURN loADO.oRecordSet
ENDFUNC

This method is fairly simple. It takes a search criteria resource (which we will be discussing in another step) as a parameter and returns an ADO record set with, presumably, the results of the search. In our example, this will be a set of consultants that match the search criteria we passed.

It’s important to note that most of this code is generic and will be used for any GetXXXBySearchCriteria() methods in adatasources.prg. The only differences will be in the method name and the select statement within the method.

Step 3: Adding A Search Method To The Consultant Class

Modify the aresourcemanagers.prg found in the ResourceMgr project. Scroll down until you find the class called Consultant and add the following method and code to the end of that class:

*---------------------------------
FUNCTION GetConsultantBySearchCriteria( tuFilter )
*---------------------------------
LOCAL loDataSource, loADORecordSet 

loDataSource = .NULL.
loADORecordSet = .NULL.
THIS.GetDataSourceObject( @loDataSource )
loADORecordSet = loDataSource.GetConsultantBySearchCriteria( tuFilter )
RETURN loADORecordSet
ENDFUNC
This code is straightforward. It simply gets an object reference to the data source class and calls the GetConsultantBySearchCriteria() method in that class. This is the method we added in the previous step. Once again, this method will return an ADO record set.

Step 4: Adding A Search Method To The ConsultantRMProxy Class

Modify the aresourcemanagerproxys.prg found in the TimeTrac project. Scroll down until you find the class called ConsultantRMProxy and add the following method and code to the end of that class:

*---------------------------------------
FUNCTION GetConsultantBySearchCriteria( tuFilter )
*---------------------------------------
LOCAL loResourceManager, lnRetVal, loADORecordSet
lnRetVal = FILE_ERROR
IF ( THIS.Get( @loResourceManager ) = FILE_OK )
loADORecordSet = ;
loResourceManager.GetConsultantBySearchCriteria( tuFilter )

IF VARTYPE( loADORecordSet ) == "O"
lnRetVal = FILE_OK
ENDIF
loResourceManager = NULL
ENDIF

IF lnRetVal = FILE_OK
THIS.oMemento.SetMethod( PROGRAM() )
THIS.oMemento.AddParameter( "tuFilter", tuFilter )
ENDIF

RETURN loADORecordSet
ENDFUNC

This should be pretty familiar by now. The code is simply making a call down to the lower level object, in this case the resource manager, and a record set is returned.

Step 5: Creating A Search Form

We now have the non-GUI side of things out of the way and are ready to start subclassing for the interface. We will create a simple search form that allows us to enter various search criteria and populates a grid with the results. First, we need to create a form. Go to the "Class" tab of the TimeTrac project and click the new button. When the "New Class" dialog pops up, enter the information as seen in Figure 1.


Figure 1

This will create a new form class named frmConsultantSearch based on the IresourceSearchForm and place the form in the aconsultant class library. You should now have the class designer open with a blank form class. 

Step 6: Adding A Proxy Loader To The Form

The first thing the form needs is a way to access data. COMCodebook accesses data in VFP via ResourceManagerProxy (RMP) classes. Remember, we’ve already added the appropriate search method to our ConsultantRMProxy, now we just have to make sure the class is instantiated along with the form. We use a ResourceManagerProxyLoader (RMPL) to do this. The TimeTrac application already has a consultant RMPL created. It can be found in the (oddly enough) armproxyloader class library as seen in Figure 2.

Figure 2

Drag the aConsultantRMProxyLoader class and drop it on the form. Now, position the class so that the left position is at 12 pixels and the top position is at 21 pixels. This is important because COMCodebook will crash unless this is positioned just so…just kidding, you can drop it anywhere on the form. With the RMPL in place the form will now be able to communicate with the ConsultantRMProxyClass that we modified in step 4.

Step 7: Adding the IResourceSearchCriteriaGrid To The Form

In order to be able to search for consultants we need a way to get a search criteria resource, add our search criteria to the resource, pass the resource to our RMP and convert the result set returned to a VFP cursor. We can do this manually or by using a subclass of the IResourceSearchCriteriaGrid to do this work for us. In this example, we will use the class to speed us along.

Find the IResourceSearchCriteriaGrid class in the iutils class library and drag it onto the form. Right-click the grid and choose "properties." The properties sheet for the grid should now be open. The beauty of this class is that the developer only needs to enter a few properties for it to work. See figure 3.

Figure 3

The important properties here are cGetBySearchCriteriaName, cResourceCursorName and cResourceManagerName. The cGetBySearchCriteriaName is the name of the method in the ConsultantRMProxy that we will use to search with. Recall in step 4 we added a method named GetConsultantBySearchCriteria(). The cResourceCursorName is the name we want to give the VFP cursor that will be populated with the search criteria. CResourceManagerName is the name of the resource manager as created by the RMPL. This can be found by examining the cRMProxyProperty property of the RMPL. In this case the property is this.parent.oConsultant. Ignore any parent references and just use the object name. Our grid class is now ready.

Step 8: Adding A Grid To Display Search Results

Drop a basic grid class onto the search form and change the name to grdSearch. Now change the RecordSource property to be a single space--this simply prevents the grid from displaying whatever cursor happens to be open at the time. This grid will be populated with the search results. In a production environment you would probably want to add appropriate columns and header captions to the grid so it sizes correctly and looks nice, but for this example we will leave it as is.

Step 9: Adding A "Begin Search" Button

The next-to-last thing we need to do is to add a button that initiates the search process. Drop a command button on the form and modify the caption property to "Begin Search". Now, open the click method and add the following code:

THISFORM.grdSearch.RECORDSOURCE = ""
IF USED("ccSearchccConsultant")
USE IN ccsearchccConsultant
ENDIF
THISFORM.IResourceSearchCriteriaGrid1.GetBySearchCriteria('ccSearch''ccConsultant')
THISFORM.grdSearch.RECORDSOURCE = "ccSearchccConsultant"
THISFORM.REFRESH()

The grid’s GetBySearchCriteria() method does all the work for us and creates a cursor named ccSearchccConsultant. We then bind this cursor to the grid to display it.

Step 10: Recompile DataSource and ResoucreMgr DLLs

Before attempting to run the form and try out our search we first need to recompile both the DataSource and the ResourceMgr projects into new DLLs. Choose the "Build" option on both projects and rebuild as a DLL. For our example it does not matter if you rebuild as a single-threaded or multi-threaded DLL.

Step 11: Search!

The moment of truth is upon us. Run the application, suspend it, and type =DoForm(‘frmConsultantSearch’) in the command window. The consultant search form should pop up. Notice that the search criteria grid has automatically populated the name and operator columns. Isn’t OOP great? Try typing a letter into the value column of the LastName row. Depending on the data in the database, your form should look something like Figure 4.

Figure 4

Try playing with the different operators and values and see what the result set looks like. If you are unfamiliar with what names are in the consultant database, simply leave all the search criteria values blank and click the begin search button. This will return all the rows from the database.

Conclusion

You should now be familiar with the mechanics of creating a VFP search form in COMCodebook. You should also have a better grasp of how the different COMCodebook tiers interact with each other and how to pass information between the tiers.

